

**Structured Computation on Trees  
or, What's Behind That Zipper? (A Comonad)**

**Tarmo Uustalu**

**TSEM, 8 Dec. 2005**

## THAT ZIPPER?

- Ever had to represent trees with a focus of attention and refocussing operations?
- The way: know or reinvent Huet's zipper (JFP 97).
- Idea: a tree with a focus = a path structure and a subtree
- Computations on trees: attribute evaluation, XML text manipulation.
- In attribute evaluation, the shape of the tree does not change and the same computation is done at every node (of the same type).
- Attribute evaluation: comonadic computation on the comonad of the zipper, similarly to dataflow computation which is also comonadic.

## OUTLINE

- Huet's zipper
- Comonads
- Attribute evaluation as comonadic computation

## ATTRIBUTED BINARY TREES

- We are interested in attributed binary trees. Mathematically,

$$\begin{aligned}\text{Tree } E A &= \mu X. A \times (E + X \times X) \\ &\cong A \times (E + \text{Tree } E A \times \text{Tree } E A)\end{aligned}$$

- In Haskell,

```
data Tree e a = a :< Trunk e (Tree e a)
data Trunk e x = Leaf e | Bin x x
```

## THE ZIPPER

- A path structure determines everything above and around a node in a hypothetical tree.

It is a path (list of turns) decorated with node attributes and side trees.

The list is a snoc-list: we want to read it left-to-right (= down from root to focus), but analyse it right-to-left (= up from focus to root).

- Mathematically,

$$\begin{aligned}\text{Cxt } E A &= \mu X. 1 + X \times (A \times \text{Tree } E A + A \times \text{Tree } E A) \\ &\cong 1 + \text{Cxt } E A \times (A \times \text{Tree } E A + A \times \text{Tree } E A)\end{aligned}$$

- In Haskell,

```
data Cxt e a = Nil | Cxt e a :> Turn a (Tree e a)
```

```
type Turn x y = Either (x, y) (x, y)
```

```
data Either z w = Left z | Right w
```

- A tree with a focus = a path structure and a (sub)tree:

$$\text{CxtTree } E A = \text{Cxt } E A \times \text{Tree } E A$$

or,

```
data CxtTree e a = Cxt e a :=| Tree e a
```

## ZIPPER NAVIGATION: DOING AND UNDOING THE ZIPPER

- Moving up (doing):

```
goParSibl :: CxtTree e a -> Maybe (Turn (CxtTree e a) (CxtTree e a))
```

```
goParSibl (Nil :=| as) = Nothing
```

```
goParSibl (az :> Left (a, asr) :=| as)
  = Just (Left (az :=| (a :< Bin as asr),
               (az :> Right (a, as) :=| asr)))
```

```
goParSibl (az :> Right (a, asl) :=| as)
  = Just (Right (az :=| (a :< Bin asl as),
                (az :> Left (a, as) :=| asl)))
```

- Moving down (undoing):

```
goChildren :: CxtTree e a -> Trunk e (CxtTree e a)
```

```
goChildren (az :=| (a :< Leaf e)) = Leaf e
```

```
goChildren (az :=| (a :< Bin asl asr))
  = Bin (az :> Left (a, asr) :=| asl)
       (az :> Right (a, asl) :=| asr)
```

## COMONADS

- Comonads are a formal dual of monads. Monads are extensively used in functional programming to structure effectful computation.
- A *comonad* is a type constructor  $D$  with two polymorphic functions  $\epsilon : D A \rightarrow A$  (the *counit*) and  $-^\dagger : (D A \rightarrow B) \rightarrow (D A \rightarrow D B)$  (the *coKleisli extension*), satisfying certain laws.
- In Haskell,

```
class Comonad d where
  counit  :: d a -> a
  cobind  :: (d a -> b) -> d a -> d b
```



- A function  $f : D A \rightarrow B$  is called a *coKleisli arrow* (of  $D$ ) from  $A \rightarrow B$  (notation  $f : A \rightsquigarrow B$ ).
- The counit is the identity coKleisli arrow, the coKleisli extension gives that they can be composed. The comonad laws guarantee that composition is associative and has the identity of as the left and right unit.
- Any function  $f : A \rightarrow B$  can be lifted to a coKleisli arrow  $Jf : A \rightsquigarrow B$ : take  $Jf = f \circ \epsilon_A$

- The Kleisli arrows of a comonad correspond to context-dependent functions.
- A context-dependent function from  $A$  to  $B$  is a function  $DA \rightarrow B$  i.e., a coKleisli map  $A \rightsquigarrow B$ .
- $DA$  is the type of contextually situated values of type  $A$ .
- The counit  $\epsilon_A : DA \rightarrow A$  discards the context of its input.
- The coextension  $k^\dagger : DA \rightarrow DB$  of a function  $k : DA \rightarrow B$  duplicates the context of its input (to feed it to  $k$  and still have a copy left).
- Examples: product comonad, comonads of dataflow computation (Uustalu, Vene, APLAS 2005)
- Examples of today: trees and zippers and attribute evaluation

## ATTRIBUTE GRAMMARS

- An attribute grammar is a CF grammar augmented with attributes and semantic equations.
- We consider a fixed CF grammar with one non-terminal  $S$  and production rules

$$\begin{aligned} S &\longrightarrow E \\ S &\longrightarrow SS \end{aligned}$$

$E$  is a pseudoterminal.

- For an attribute type  $A$  for  $S$ -nodes, the type of attributed  $S$ -trees is  $\text{Tree } E A$ .

## PURELY SYNTHESIZED ATTRIBUTE GRAMMARS

- For a purely synthesized attribute grammar, the local value of a defined attribute of a semantic equation can explicitly depend on the local and children-node values of the defining attributes.
- The relevant comonad is  $\text{Tree } E$ , where the value under focus is the root attribute value and those in the subtrees form are its context.
- In Haskell,

```
instance Comonad (Tree e) where
  counit (a :< _) = a
  cobind k d@(_ :< as) = k d :< case as of
    Leaf e      -> Leaf e
    Bin asL asR -> Bin (cobind k asL) (cobind k asR)
```

- The Kleisli arrows represent tree functions:

```
class TF e d where
  run :: (d e a -> b) -> Tree e a -> Tree e b
```

```
instance TF e Tree where
  run = cobind
```

- Looking up the values at the children:

```
class Synth e d where
  children :: d e a -> Trunk e a
```

```
instance Synth e Tree where
  children (_ :< as) = case as of
    Leaf e                -> Leaf e
    Bin (aL :< _) (aR :< _) -> Bin aL aR
```

## GENERAL ATTRIBUTE GRAMMARS

- Here we need a more permissive notion of context. This is provided by the zipper.
- Now `CxtTree E` is a comonad as well!

```
instance Comonad (CxtTree e) where
```

```
  counit (_ :=| (a :< _)) = a
```

```
  cobind k d = cobindC k d :=| cobindT k d
```

```
  where cobindC k d = case goParSibl d of
```

```
    Nothing -> Nil
```

```
    Just (Left (d', dR)) ->
```

```
        cobindC k d' :> Left (k d', cobindT k dR)
```

```
    Just (Right (d', dL)) ->
```

```
        cobindC k d' :> Right (k d', cobindT k dL)
```

```
  cobindT k d = k d :< case goChildren d of
```

```
    Leaf e -> Leaf e
```

```
    Bin dL dR -> Bin (cobindT k dL) (cobindT k dR)
```

- CoKleisli arrows can be interpreted as tree functions and there is an operation for getting the values of the attribute at the children of a node.

```
instance TF e CxtTree where
```

```
  run k as = bs where Nil :=| bs = cobind k (Nil :=| as)
```

```
instance Synth e CxtTree where
```

```
  children (_ :=| (_ :< as)) = case as of
```

```
    Leaf e                               -> Leaf e
```

```
    Bin (aL :< _) (aR :< _) -> Bin aL aR
```

- There is also an operation that checks whether the local node is the root or not, and if it is not, obtains the value of the attribute at the parent and the sibling.

```
class Inh e d where
```

```
  parSibl :: d e a -> Maybe (Either (a, a) (a, a))
```

```
instance Inh e CxtTree where
```

```
  parSibl (Nil      :=| _) = Nothing
```

```
  parSibl (_ :=> b :=| _) = Just $ case b of
```

```
    Left  (a, a' :< _) -> Left  (a, a')
```

```
    Right (a, a' :< _) -> Right (a, a')
```



## EXAMPLES

- Checking a tree for AVLness.
- Attribute grammar presentation:

$$S^\ell \longrightarrow E$$

$$S^b \longrightarrow S_L^b S_R^b$$

$$S^\ell.avl = tt$$

$$S^b.avl = S_L^b.avl \wedge S_R^b.avl \wedge S^b.locavl$$

$$S^\ell.locavl = tt$$

$$S^b.locavl = |S_L^b.height - S_R^b.height| \leq 1$$

$$S^\ell.height = 0$$

$$S^b.height = \max(S_L^b.height, S_R^b.height) + 1$$

- Comonadic Haskell:

```
avl :: (Comonad (d e), Synth e d) => d e () -> Bool
```

```
avl d = case children d of
```

```
    Leaf _ -> True
```

```
    Bin _ _ -> bL && bR && locavl d
```

```
        where Bin bL bR = children (cobind avl d)
```

```
locavl :: (Comonad (d e), Synth e d) => d e () -> Bool
```

```
locavl d = case children d of
```

```
    Leaf _ -> True
```

```
    Bin _ _ -> abs (hL - hR) <= 1
```

```
        where Bin hL hR = children (cobind height d)
```

```
height :: (Comonad (d e), Synth e d) => d e () -> Integer
```

```
height d = case children d of
```

```
    Leaf _ -> 0
```

```
    Bin _ _ -> max hL hR + 1
```

```
        where Bin hL hR = children (cobind height d)
```

- Preorder numbering of the nodes of a tree.
- Attribute grammar presentation:

$$S^\ell \longrightarrow E$$

$$S^b \longrightarrow S_L^b S_R^b$$

$$S_L^b.\textit{numin} = S^b.\textit{numin} + 1$$

$$S_R^b.\textit{numin} = S_L^b.\textit{numout} + 1$$

$$S^\ell.\textit{numout} = S^\ell.\textit{numin}$$

$$S^b.\textit{numout} = S_R^b.\textit{numout}$$

- Comonadic Haskell:

```
numin :: (Comonad (d e), Synth e d, Inh e d) => d e () -> Integer
numin d = case parSibl d of
  Nothing -> 0
  Just (Left _) -> ni + 1
    where Just (Left (ni, _)) = parSibl (cobind numin d)
  Just (Right _) -> noL + 1
    where Just (Right (_, noL)) = parSibl (cobind numout d)

numout :: (Comonad (d e), Synth e d, Inh e d) => d e () -> Integer
numout d = case children d of
  Leaf e -> numin d
  Bin _ _ -> noR
    where Bin _ noR = children (cobind numout d)
```

## CONCLUSIONS

- The zipper datatype hides a comonad. This is exactly the comonad one needs to structure attribute evaluation.
- GADTs+typecase or dependent types needed to smoothly deal with multiple non-terminals, multiple rules for individual non-terminals.
- To do: Comonadically interpreted AG definition language.